Econ 741 (Spring 2003):

Suggested Answer for Computer Homework 1

```
        @ A. COMPUTER EXERCISE @

        @ PROGRAM @
new;
output file = c:\E741\nerlove.out reset;

        @ read a data file @
load hwdata[145,5]=c:\E741\nerlove.dat;

        @ transform data @
hwdata = ln(hwdata);

        @ assigning each column to each variable @
tc = hwdata[.,1]; @ assigning 1st col to Total Cost @
q = hwdata[.,2]; @ assigning 2nd col to Output @
p1 = hwdata[.,3]; @ assigning 3rd col to Wage rate @
p2 = hwdata[.,5]; @ assigning 4th col to Price of fuel @
p3 = hwdata[.,4]; @ assigning 5th col to rental price @

        @ Unrestricted model equation (1.7.4) @
"1. Unrestricted Model";
"log(tc) = b1 + b2 log(q) + b3 log(p1) + b4 log(p2) + b5 log(p3) + e";
"";

        @ define regressand and regressors @
y = tc;
x = ones(145,1)~q~p1~p2~p3;
n = rows(y);
k = cols(x);
dfu = n - k; @ degree of freedom for unrestricted model @

        @ obtain OLS estimates @
ixx = invpd(x'x);
bhat = ixx*x'y;

        @ obtain SSR(sum of squared residuals) and R2 @
ehat = y-x*bhat; @ residual @
ssr = ehat'ehat; @ sum of squared residuals @
tss = (y-meanc(y))'(y-meanc(y)); @ total sum of squares @
r2 = 1 - ssr/tss; @ the coefficients of determination @
```

```
        @ obtain standard errors of OLS estimates @
sigmahat = ssr / dfu;
covbhat = sigmahat*ixx; @ covariance matrix of OLS estimates @
stdbhat = sqrt(diag(covbhat)); @ standard errors of OLS estimates @

        @ get t-values and p-values of OLS estimates @
tvbhat = bhat./stdbhat; @ element by element division @
pvalue = 2*cdftc(abs(tvbhat),dfu); @ for two tail test @

        @ print results @
let vnames[5,1] = constant Q P1 P2 P3;
"-------------------------------------------------------------";
" Variable     Coefficient     Std.Error     t-ratio     p-value ";
"-------------------------------------------------------------";
format /rdn 13,4;
i=1;
do while i <= k;
$vnames[i];;bhat[i];;stdbhat[i];;tvbhat[i];;pvalue[i];
i=i+1;
endo;
"-------------------------------------------------------------";
"";
print "Sum of squared residuals: " ssr;
print "R-square (centered): " r2;

        @ restricted Model: equation (1.7.6) @
"2. Restricted Model";
"log(tc) = b1 + b2 log(q) + b3 log(p1) + b4 log(p2) + (1-b3-b4)log(p3) + e";
"";

        @ define regressand and regressors @
yr = tc-p3;
xr = ones(145,1)~q~(p1-p3)~(p2-p3);
nr = rows(yr);
kr = cols(xr);
dfr = nr - kr;

        @ obtain OLS estimates @
ixxr = invpd(xr'xr);
bhatr = ixxr*xr'yr;

        @ obtain SSR(sum of squared residuals) and R2 @
ehatr = yr-xr*bhatr;
ssrr = ehatr'ehatr;
tssr = (yr-meanc(yr))'(yr-meanc(yr));
```

```
r2r = 1 - ssrr/tssr;

        @ obtain standard errors of OLS estimates @
sigmar = ssrr/ dfr;
covbhatr = sigmar*ixxr;
stdbhatr = sqrt(diag(covbhatr));

        @ get t-values and p-values of OLS estimates @
tvbhatr = bhatr./stdbhatr;
pvaluer = 2*cdftc(abs(tvbhatr),dfr);

        @ print results @
let vnames[4,1] = constant Q P1/P3 P2/P3;
"_____";
" Variable     Coefficient     Std.Error     t-ratio     p-value ";
"_____";
format /rdn 13,4;
i=1;
do while i<=kr;
$ vnames[i];;bhatr[i];;stdbhatr[i];;tvbhatr[i];;pvaluer[i];
i=i+1;
endo;
"_____";
"";
print "Sum of squared residuals: " ssrr;
print "R-square (centered): " r2r;

        @ F-test for Homogeneity of the Cost Function @
"3. F-test for Homogeneity of the Cost Function (H0:b3+b4+b5=1): ";
"Using (1.4.11)";
q = 1;
dfu = dfu;
fstath = ((ssrr-ssr)/q)/(ssr/dfu);
pvalueh = cdffc(fstath,q,dfu);
format /rdn 1,0;
"F(";;q;;", ";;dfu;;" ) = ";;
format /rdn 13,4;
fstath;
"p-value = ";; pvalueh;
"_____";
"";

        @ F-test for the joint hypothesis (H0: b2=b3=b4=b5=0)
: To test this, first you have to estimate resricted model
under the null hypothesis @
```

```
"4. F-test of H0: b2=b3=b4=b5=0 ";
"4-1). Restricted Model";
"log(tc) = b1 + e";
"";

        @ define regressand and regressors @
yr2 = tc;
xr2 = ones(145,1);
nr2 = rows(yr2);
kr2 = cols(xr2);
dfr2 = nr2 - kr2;

        @ obtain OLS estimates @
ixxr2 = invpd(xr2'xr2);
bhatr2 = ixxr2*xr2'yr2;

        @ obtain SSR(sum of squared residuals) and R2 @
ehatr2 = yr2-xr2*bhatr2;
ssrr2 = ehatr2'ehatr2;
tssr2 = (yr2-meanc(yr2))'(yr2-meanc(yr2));
r2r2 = 1 - ssrr2/tssr2;

        @ obtain standard errors of OLS estimates @
sigmar2 = ssrr2/ dfr2;
covr2 = sigmar2*ixxr2;
stdr2 = sqrt(diag(covr2));

        @ get t-values and p-values of OLS estimates @
thatr2 = bhatr2./stdr2;
pvalr2 = 2*cdftc(abs(thatr2),dfr2);

        @ print results @
let vnames[1,1] = constant;
"-------------------------------------------------------------";
" Variable     Coefficient     Std.Error     t-ratio     p-value ";
"-------------------------------------------------------------";
format /rdn 13,4;
i=1;
do while i<=kr2;
$ vnames[i];;bhatr2[i];;stdr2[i];;thatr2[i];;pvalr2[i];
i=i+1;
endo;
"-------------------------------------------------------------";
"";
print "Sum of squared residuals: " ssrr2;
```

```
print "R-square (centered): " r2r2;

      "4-2) calculate F-test statistic";
"Using (1.4.11) formula";
qr = 4;
dfu = dfu;
fstatr = ((ssrr2-ssr)/qr)/(ssr/dfu);
pvaluer = cdffc(fstatr,qr,dfu);
format /rdn 1,0;
"F(";;qr;;", ";;dfu;;" ) = ";;
format /rdn 13,4;
fstatr;
"p-value = ";; pvaluer;
"_____";
"";

      "4-2)' calculate F-test statistic";
"Using simplified (1.4.11) formula in problem sheet";
fstatr2 = r2/(1-r2) * dfu/qr;
pvaluer2 = cdffc(fstatr2,qr,dfu);
format /rdn 1,0;
"F(";;qr;;", ";;dfu;;" ) = ";;
format /rdn 13,4;
fstatr2;
"p-value = ";; pvaluer2;
"_____";
"";
end;
```

@ OUTPUT @

1. Unrestricted Model

log(tc) = b1 + b2 log(q) + b3 log(p1) + b4 log(p2) + b5 log(p3) + e

Variable	Coefficient	Std.Error	t-ratio	p-value
CONSTANT	-3.5265	1.7744	-1.9875	0.0488
Q	0.7204	0.0175	41.2445	0.0000
P1	0.4363	0.2910	1.4992	0.1361
P2	-0.2199	0.3394	-0.6478	0.5182
P3	0.4265	0.1004	4.2495	0.0000

Sum of squared residuals: 21.5520
R-square (centered): 0.9260

2. Restricted Model
log(tc) = b1 + b2 log(q) + b3 log(p1) + b4 log(p2) + (1-b3-b4)log(p3) + e

Variable	Coefficient	Std.Error	t-ratio	p-value
CONSTANT	-4.6908	0.8849	-5.3011	0.0000
Q	0.7207	0.0174	41.3340	0.0000
P1/P3	0.5929	0.2046	2.8983	0.0044
P2/P3	-0.0074	0.1907	-0.0387	0.9692

Sum of squared residuals: 21.6403,
R-square (centered): 0.9316

3. F-test for Homogeneity of the Cost Function (H0:b3+b4+b5=1): Using (1.4.11)
F(1, 140) = 0.5737
p-value = 0.4501

4. F-test of H0: b2=b3=b4=b5=0
4-1). Restricted Model, log(tc) = b1 + e

Variable	Coefficient	Std.Error	t-ratio	p-value
CONSTANT	1.7247	0.1181	14.6074	0.0000

Sum of squared residuals: 291.0668,
R-square (centered): 0.0000

4-2) calculate F-test statistic Using (1.4.11) formula
F(4, 140) = 437.6863
p-value = 0.0000

4-2)' calculate F-test statistic Using simplified (1.4.11) formula in problem sheet
F(4, 140) = 437.6863
p-value = 0.0000

@ B.Exercises @

1) 95% confidence interval for β_2 in (1.7.6): Let t_c be a critical value. Then the interval is $b_2 - SE(b_2) \cdot t_c \leq \beta_2 \leq b_2 + SE(b_2) \cdot t_c$. Substitute $b_2 = 0.7207, SE(b_2) = 0.0174, t_c = 1.98$. Then, the confidence interval of β_2 is $0.6862 \leq \beta_2 \leq 0.7552$

2) The t-value in (1.7.6) is 2.8983 which is greater than $t_c = 1.98$, so we can reject the null $H_0 : \beta_3 = 0$

3) The p-value of the test is 0.4501 which is larger than the significance level (5%). So, we can not reject the null $H_0 : \beta_1 + \beta_2 + \beta_3 = 1$

4) The p-value of the test is 0.0000 which is smaller than the significance level (1%). So, we can reject the null $H_0 : \beta_2 = \beta_3 = \beta_4 = 0$